Всем доброе времени суток! Столкнулся, казалось бы с банальной задачей: Модифицировать запрос формы списка заказов клиента (УТ 11.4), добавить левым соединением регистр сведений по ссылке заказа, выбрать из него пару ресурсов. Установить условное оформление (цвет фона всей таблицы) на значение ресурса... Но ничего не работает, т.е. эффекта нету. Стал экспериментировать. Если установить отбор на существующее поле (которое уже было), то УО работает, а вот на новое поле нет. Текст запроса получается как нужно. Ошибок ни каких не пишет. Модификацию всея выполняю в МодификацияКонфигурацииПереопределяемый.ПриСозданииНаСервере в расширении. Платформа 8.3.15.1656.

```1c
    СхемаЗапроса = Новый СхемаЗапроса;
    СхемаЗапроса.УстановитьТекстЗапроса(Форма.Список.ТекстЗапроса);
    Оператор = СхемаЗапроса.ПакетЗапросов[0].Операторы[0];

    ИсточникПротокол = Оператор.Источники.Добавить("РегистрСведений.др_ПротоколОбменаССайтом.СрезПоследних", "ПротоколОбменаССайтом");
    ИсточникПротокол.Соединения[0].ТипСоединения = ТипСоединенияСхемыЗапроса.ПравоеВнешнее;
    ИсточникПротокол.Соединения[0].Условие = Новый ВыражениеСхемыЗапроса("ДокументЗаказКлиента.Ссылка = ПротоколОбменаССайтом.Объект");

    ПолеСтатусЗагрузки = Оператор.ВыбираемыеПоля.Добавить("ПротоколОбменаССайтом.СтатусЗагрузки");
    СхемаЗапроса.ПакетЗапросов[0].Колонки[СхемаЗапроса.ПакетЗапросов[0].Колонки.Количество() - 1].Псевдоним = "СтатусЗагрузки";

    ПолеТекстСообщения = Оператор.ВыбираемыеПоля.Добавить("ПротоколОбменаССайтом.ТекстСообщения");
    СхемаЗапроса.ПакетЗапросов[0].Колонки[СхемаЗапроса.ПакетЗапросов[0].Колонки.Количество() - 1].Псевдоним = "ТекстСообщения";

    ТекстЗапроса = СхемаЗапроса.ПолучитьТекстЗапроса();
    Форма.Список.ТекстЗапроса = ТекстЗапроса;

    //

    //ЭлементОформления = Форма.Список.КомпоновщикНастроек.Настройки.УсловноеОформление.Элементы.Добавить();
    //ЭлементОформления.Представление = НСтр("ru = 'Выделение цветом заказов, загруженные с ошибкой'");
    ЭлементОформления = Форма.УсловноеОформление.Элементы.Добавить();

    ПолеОформления = ЭлементОформления.Поля.Элементы.Добавить();
    ПолеОформления.Поле = Новый ПолеКомпоновкиДанных("Список");

    ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбора(ЭлементОформления.Отбор, "Список.СтатусЗагрузки", Перечисления.др_СтатусыОбменаССайтом.Ошибка);
    //ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбора(ЭлементОформления.Отбор, "Список.Статус", Перечисления.СтатусыЗаказовКлиентов.КОбеспечению);

    ЭлементОформления.Оформление.УстановитьЗначениеПараметра("ЦветФона", WebЦвета.Коралловый);
```
---
Всем спасибо. Проблема была в пользовательских настройках пользователя. Видать какая-то особенность работы платформы. Чтобы все заработало, нужно сбросить их. Вот итоговый рабочий кот:

```1c
    СхемаЗапроса = Новый СхемаЗапроса;
    СхемаЗапроса.УстановитьТекстЗапроса(Форма.Список.ТекстЗапроса);
    Оператор = СхемаЗапроса.ПакетЗапросов[0].Операторы[0];

    ИсточникПротокол = Оператор.Источники.Добавить("РегистрСведений.др_ПротоколОбменаССайтом.СрезПоследних", "ПротоколОбменаССайтом");
    ИсточникПротокол.Соединения[0].ТипСоединения = ТипСоединенияСхемыЗапроса.ПравоеВнешнее;
    ИсточникПротокол.Соединения[0].Условие = Новый ВыражениеСхемыЗапроса("ДокументЗаказКлиента.Ссылка = ПротоколОбменаССайтом.Объект");

    Оператор.ВыбираемыеПоля.Добавить("ПротоколОбменаССайтом.СтатусЗагрузки");
    Оператор.ВыбираемыеПоля.Добавить("ПротоколОбменаССайтом.ТекстСообщения");

    ТекстЗапроса = СхемаЗапроса.ПолучитьТекстЗапроса();
    Форма.Список.ТекстЗапроса = ТекстЗапроса;

    ЭлементОформления = Форма.Список.КомпоновщикНастроек.Настройки.УсловноеОформление.Элементы.Добавить();

    ЭлементОформления.Представление = НСтр("ru = 'Выделение цветом заказов, загруженные с ошибкой'");
    ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбора(ЭлементОформления.Отбор, "СтатусЗагрузки", Перечисления.др_СтатусыОбменаССайтом.Ошибка);

    ЭлементОформления.Оформление.УстановитьЗначениеПараметра("ЦветФона", Новый Цвет(255, 200, 200));

    //

    ЭлементОформления = Форма.Список.КомпоновщикНастроек.Настройки.УсловноеОформление.Элементы.Добавить();
    ЭлементОформления.Представление = НСтр("ru = 'Выделение цветом заказов, загруженные с предупреждением'");

    ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбора(ЭлементОформления.Отбор, "СтатусЗагрузки", Перечисления.др_СтатусыОбменаССайтом.Предупреждение);

    ЭлементОформления.Оформление.УстановитьЗначениеПараметра("ЦветФона", WebЦвета.СеребристоСерый);
```
---
Всем спасибо. Проблема была в пользовательских настройках пользователя. Видать какая-то особенность работы платформы. Чтобы все заработало, нужно сбросить их. Вот итоговый рабочий кот:

```1c
    СхемаЗапроса = Новый СхемаЗапроса;
    СхемаЗапроса.УстановитьТекстЗапроса(Форма.Список.ТекстЗапроса);
    Оператор = СхемаЗапроса.ПакетЗапросов[0].Операторы[0];

    ИсточникПротокол = Оператор.Источники.Добавить("РегистрСведений.др_ПротоколОбменаССайтом.СрезПоследних", "ПротоколОбменаССайтом");
    ИсточникПротокол.Соединения[0].ТипСоединения = ТипСоединенияСхемыЗапроса.ПравоеВнешнее;
    ИсточникПротокол.Соединения[0].Условие = Новый ВыражениеСхемыЗапроса("ДокументЗаказКлиента.Ссылка = ПротоколОбменаССайтом.Объект");

    Оператор.ВыбираемыеПоля.Добавить("ПротоколОбменаССайтом.СтатусЗагрузки");
    Оператор.ВыбираемыеПоля.Добавить("ПротоколОбменаССайтом.ТекстСообщения");

    ТекстЗапроса = СхемаЗапроса.ПолучитьТекстЗапроса();
    Форма.Список.ТекстЗапроса = ТекстЗапроса;

    ЭлементОформления = Форма.Список.КомпоновщикНастроек.Настройки.УсловноеОформление.Элементы.Добавить();

    ЭлементОформления.Представление = НСтр("ru = 'Выделение цветом заказов, загруженные с ошибкой'");
    ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбора(ЭлементОформления.Отбор, "СтатусЗагрузки", Перечисления.др_СтатусыОбменаССайтом.Ошибка);

    ЭлементОформления.Оформление.УстановитьЗначениеПараметра("ЦветФона", Новый Цвет(255, 200, 200));

    //

    ЭлементОформления = Форма.Список.КомпоновщикНастроек.Настройки.УсловноеОформление.Элементы.Добавить();
    ЭлементОформления.Представление = НСтр("ru = 'Выделение цветом заказов, загруженные с предупреждением'");

    ОбщегоНазначенияКлиентСервер.УстановитьЭлементОтбора(ЭлементОформления.Отбор, "СтатусЗагрузки", Перечисления.др_СтатусыОбменаССайтом.Предупреждение);

    ЭлементОформления.Оформление.УстановитьЗначениеПараметра("ЦветФона", WebЦвета.СеребристоСерый);
```